# Maintainer: Brett Cornwall <ainola@archlinux.org>
# Maintainer: Maxim Baz <$pkgname at maximbaz dot com>

pkgname=urlscan
pkgver=0.9.10
pkgrel=2
pkgdesc="Mutt and terminal url selector"
url="https://github.com/firecat53/urlscan"
arch=('any')
license=('GPL2')
depends=('python-urwid')
makedepends=('python-setuptools')
optdepends=(
    'xclip: Clipboard support for X11 environments'
    'wl-clipboard: Clipboard support for Wayland'
)
source=("$pkgname-$pkgver.tar.gz::https://github.com/firecat53/urlscan/archive/$pkgver.tar.gz")
sha256sums=('0fbaf7f820fc6ff20c6f1a429c98cb240f1da7eabceed5e3d88740558e81cd2d')

build() {
    cd "$pkgname-$pkgver"
    python setup.py build
}

package() {
    cd "$pkgname-$pkgver"
    python setup.py install --root="$pkgdir/" --optimize=1 --skip-build
}
